

Fedlex-JOLux documentation

Release 0.1

Jean-Louis Morard, Benedikt Hitz-Gamper

Dec 11, 2024

List of Figures

iii

1	Introduction	1
1.1	Fedlex	1
1.2	JOLux Ontology	1
1.3	How to Use this Website	1
1.4	Website as PDF	2
1.5	SPARQL Queries	2
1.6	SPARQL Tutorial for JOLux	3
1.7	Fedlex URI and URL	3
1.8	Namespaces Declarations	3
2	Abstraction Levels	5
2.1	Work, Expression and Manifestation.	5
2.2	Object Properties	5
2.3	SPARQL Examples	7
3	Official Compilation (OC)	9
3.1	Example	9
3.2	URI	9
3.3	General Structure	10
3.4	Datatype Properties	12
3.5	Object Properties	12
3.6	SPARQL Examples	12
4	Classified Compilation (CC)	15
4.1	Example	15
4.2	URI	15
4.3	General Structure	16
4.4	Legal Taxonomy	17
4.5	Datatype Properties	17
4.6	Object Properties	18
4.7	SPARQL Examples	18
5	Impacts	21
5.1	Example	21
5.2	URI	21
5.3	General Structure	21
5.4	Datatype Properties	23
5.5	Object Properties	23
5.6	SPARQL Examples	23

6	Subdivisions	25
6.1	Example	25
6.2	URI	25
6.3	General Structure	25
6.4	Object Properties	26
6.5	SPARQL Examples	27
7	History	29
7.1	Draft	29
7.2	Impact	30
8	Changes	33
8.1	Older Changes	33
8.2	Newer Changes	33
8.3	SPARQL Examples	33
9	Citations	35
9.1	From and To	37
9.2	SPARQL Example	37
10	International Treaties	39
10.1	Example	39
10.2	URI	39
10.3	General Structure	39
10.4	Datatype Properties for jolux:TreatyProcess	42
10.5	Object Properties for jolux:TreatyProcess	42
10.6	SPARQL Examples	42
11	Vocabularies	45
11.1	Available Vocabularies	45
11.2	Hierarchical Vocabularies	46
11.3	Act Types.	46
11.4	Countries	47
11.5	Enforcement Status	47
11.6	Impact Types	48
11.7	Legal Institution.	48
11.8	Legal Taxonomy.	49
11.9	Procedure Types	50
11.10	Subdivision Types	50
11.11	Text Types	51
11.12	Treaty Subject Themes	52
11.13	Treaty Types	52
11.14	User Formats	53
12	Dates	55
12.1	Special Dates	56
13	Reference	57
13.1	JOLux Classes	57
13.2	JOLux Datatype Properties	57
13.3	JOLux Object Properties	58
13.4	Concepts	59

Figure 1.1: Graphical representation of JOLux ontology elements.	2
Figure 2.1: Relation between jolux:Work, jolux:Expression und jolux:Manifestation.	6
Figure 3.1: General structure of an entry in the Official Compilation.	11
Figure 4.1: General structure of an entry in the Classified Compilation.	17
Figure 5.1: Connection between entries in the Official Compilation and the Classified Compilation through impacts.	22
Figure 6.1: General structure of a jolux:LegalResourceSubdivision.. . . .	26
Figure 7.1: General structure of a jolux:Draft.	30
Figure 9.1: General structure of a citation.	36
Figure 10.1: General structure of a treaty process.. . . .	40

Introduction

This introduction explains some key terms to understand the scope of its content and describes the structure of the webpage and how to use it.

1.1 Fedlex

The Swiss federal government operates the **Fedlex** platform to publish the **federal law**. This platform provides a [website](#) as frontend with easy navigable functions. For some cases, it is beneficial to work directly with the raw data that is also the basis for the frontend website. This raw data is available in [RDF](#) format through a [SPARQL GUI](#) and a SPARQL endpoint at `https://fedlex.data.admin.ch/sparqlendpoint`.

1.2 JOLux Ontology

The raw data of the Fedlex platform in RDF is modelled according to the **JOLux ontology**. This ontology is used for describing **legislative resources and their relationships**.

JOLux is based on recent developments in bibliographical description, adapting the [FRBR model](#) (Functional requirements for Bibliographic Records, developed by the [IFLA](#)) in order to describe legislative resources.

This website's goal is to document the JOLux ontology and help users to find their way into the RDF data of the Fedlex platform and make the most use of it. It is not the basis for the JOLux ontology meaning that there is no completeness of all the aspects of JOLux in this documentation. So this website can not be used to model data according to the JOLux ontology but rather to understand data that is already modelled with help of JOLux. If complete insight into the JOLux ontology is necessary, it can be [downloaded](#) as Turtle file for further investigation.

1.3 How to Use this Website

This website has sub-pages for all the important [Section 13.4](#). A concept is loosely defined an important element of the JOLux ontology. These sub-pages describe the concept in prose. Additional call-out boxes give short definition of JOLux and other terms (see the example below for [ontology](#)). These boxes are all linked in the [reference](#).

Ontology

An ontology is a set of precise descriptive statements about some part of the world (usually referred to as the domain of interest or the subject matter of the ontology). Precise descriptions satisfy several purposes: most notably, they prevent misunderstandings in human communication and they ensure

that software behaves in a uniform, predictable way and works well with other software. [Source](#)

The visual representation of parts of the JOLux ontology on this website is loosely based on the [VOWL](#) project. In addition, multiple colors represent the different [abstraction levels](#) of JOLux.

The following figure shows the elements of graphical representation of JOLux in this documentation using an example of a `jolux:Act`:




Figure 1.1. Graphical representation of JOLux ontology elements.

The figure above can be read as: Some object of type `jolux:Act` is connected to an object of type `jolux:Expression` via predicate `jolux:isRealizedBy`. So the single bubbles do not represent concrete objects but signal class memberships.

1.4 Website as PDF

This website is also available as [PDF](#).

1.5 SPARQL Queries

Throughout this webpage, there are examples of SPARQL queries given. The idea is, that these are real queries that can be executed on the [Fedlex SPARQL GUI](#) to get real up to date results. To do so,

below every SPARQL example query, there is a “Execute Query” button that transfers the corresponding query into the SPARQL GUI and executes it to show the tabular result. As the source code of these queries is also given, the user should be encouraged to modify these queries directly in the SPARQL GUI or use it programmatically to their own needs.

The following SPARQL query shows this method by giving the 10 newest published `jolux:Act` that are available:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {
  ?act a jolux:Act;
       jolux:publicationDate ?date.
} ORDER BY DESC(?date)
LIMIT 10
```

1.6 SPARQL Tutorial for JOLux

There is also a tutorial in the form of a *JupyterLite notebook* available that can be run in the browser. The main focus of this interactive tutorial are the SPARQL queries necessary to work with the Fedlex JOLux data. In the notebook, these queries can be directly executed and changed by the user. The tutorial is available [here](#).

1.7 Fedlex URI and URL

All URI of Fedlex raw data resources start with: `https://fedlex.data.admin.ch/eli` whereas `eli` is an abbreviation for [European Legislation Identifier](#).

These URI can be found on the website of [Fedlex](#) through a search. The raw data URI is not the URL shown in the browser address field but can be copied by clicking on the chain icon. If an URI is put into the browser address field, there is an automatic redirection to the webpage URL that displays the corresponding resource.

Examples for the federal constitution in the Classified Compilation:

- URI: `https://fedlex.data.admin.ch/eli/cc/1999/404`
- URL: `https://www.fedlex.admin.ch/eli/cc/1999/404`

The easiest way to have a graph like representation of a Fedlex URI (and not a redirection to the URL) is to put it into the [metadata viewer](#) of the Fedlex platform. Links to the metadata viewer with prefilled URI can also be programmatically created via URL parameter `value` with the desired URI:

```
https://fedlex.data.admin.ch/en-CH/metadata?value=https://fedlex.data.admin.ch/eli/cc/1999/404
```

The URLs given in this documentation are given without language identifier. In reality, there is no such URL as `https://www.fedlex.admin.ch/eli/cc/1999/404` but only `https://www.fedlex.admin.ch/eli/cc/1999/404/en` or with other language identifiers like `de`, `fr`, `it` or `rm` in the end. But there is a redirection mechanism in place that automatically redirects to the correct language URL according to browser settings if no language identifier is given.

1.8 Namespaces Declarations

The following namespaces are used throughout this documentation:

PREFIX	URI
jolux	http://data.legilux.public.lu/resource/ontology/jolux#
schema	http://schema.org/
skos	http://www.w3.org/2004/02/skos/core#
dcterm	http://purl.org/dc/terms/
xsd	http://www.w3.org/2001/XMLSchema#
rdfs	http://www.w3.org/2000/01/rdf-schema#
rdf	http://www.w3.org/1999/02/22-rdf-syntax-ns#
owl	http://www.w3.org/2002/07/owl#
eu	http://publications.europa.eu/resource/authority/

Abstraction Levels

In JOLux, all the different legislative resources are usually described through **three different levels of abstraction**. These levels are necessary to be able to reference legislative resources either depending on language and file format or not.

2.1 Work, Expression and Manifestation

jolux:Work

The owl:Class **jolux:Work** is a general abstraction for all the different legislative resources in JOLux. All the objects with type **jolux:Work** have additional types added to differentiate between the diverse legislative resources.

As **jolux:Work** is a general abstraction, the **jolux:Work** is *language and file-format agnostic*.

jolux:Expression

The owl:Class **jolux:Expression** is a *language specific* representation of a **jolux:Work**. The **jolux:Expression** is *file-format agnostic*.

jolux:Manifestation

The owl:Class **jolux:Manifestation** is a *file-format specific* representation of a **jolux:Expression** entity. So an **jolux:Manifestation** is a *language and file-format specific* representation of a **jolux:Work**.

So basically, **jolux:Work**, **jolux:Expression** and **jolux:Manifestation** always come together to form a rich representation of a legislative resource.

2.2 Object Properties

The vocabulary used to connect these abstraction levels is as following:

jolux:isRealizedBy

The object property **jolux:isRealizedBy** points from a **jolux:Work** to a **jolux:Expression**.

jolux:isEmbodiedBy

The object property **jolux:isEmbodiedBy** points from a **jolux:Expression** to a **jolux:Manifestation**.

jolux:isExemplifiedBy

The object property `jolux:isExemplifiedBy` points from a `jolux:Manifestation` to an object that represent the URL of the actual document.

The following figure shows the different abstraction levels and the object properties to connect them:




Figure 2.1. Relation between `jolux:Work`, `jolux:Expression` und `jolux:Manifestation`.

2.3 SPARQL Examples

The following question uses the above introduced abstraction levels to drill down from the URI of the Federal Constitution in the [Official Compilation](#) to the link of the PDF document in German:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {
  <https://fedlex.data.admin.ch/eli/oc/1999/404> jolux:isRealizedBy ?expression.
  ?expression jolux:language <http://publications.europa.eu/resource/authority
/language/DEU>;
  jolux:isEmbodiedBy ?manifestation.
  ?manifestation jolux:userFormat <https://fedlex.data.admin.ch/vocabulary
/user-format/pdf-a>;
  jolux:isExemplifiedBy ?url.
}
```

Official Compilation (OC)

The *Official Compilation* (OC) is according to the [lexicon of parliamentary terms](#) the compilation of primarily the federal constitution, federal acts and federal decrees.

This part explains all the important objects that build an entry in the Official Compilation and it does so with the help of the federal constitution as an example of an entry in the Official Compilation.

Hint for legal laypersons

Entries in the Official Compilation do not represent something like a current consolidated version of a legislative resource but are some kind of “building blocks” of an actual legislative resource. Updates to a legal text are published as “deltas” to already existing texts - much like an additional commit in software development.

In distinction from the Official Compilation, the current consolidated legislative resources are modelled in the [Classified Compilation](#).

3.1 Example

Throughout this sub-page, the federal constitution is used as an example of an entry in the Official Compilation.

- URI: <https://fedlex.data.admin.ch/eli/oc/1999/404>
- URL: <https://www.fedlex.admin.ch/eli/oc/1999/404>
- [Metadata viewer](#)

3.2 URI

The URI of an entry in the Official Compilation contains the following parts:

- Standard namespace and path: `https://fedlex.data.admin.ch/eli/`
- the part `oc/` denotes the Official Compilation, meaning that this URI identifies something that is part of the Official Compilation of the federal law
- `YYYY/` is the year of the publication (for older acts this can just be a number from 1 to 63 and roman I to XI)
- ID an identifier that has no specific meaning but is restarting every new year and for older entries, this can also be linked to the starting page number for each language (e.g. `https://fedlex.data.admin.ch/eli/oc/1/183_154_179`)

3.3 General Structure

Every entry in the Official Compilation is of type `jolux:Act`.

`jolux:Act`

The owl:Class **`jolux:Act`** is used for entries in the Official Compilation and the Federal Gazette. It is of the same **abstraction level** as `jolux:Work` and all `jolux:Act` are also `jolux:Work`.

For `jolux:Act`, the additional **abstraction levels** `jolux:Expression` and `jolux:Manifestation` are also available for all entries except some older ones (e.g. https://fedlex.data.admin.ch/eli/oc/1/183_154_179).

The following figure shows the general structure of an entry in the Official Compilation:




Figure 3.1. General structure of an entry in the Official Compilation.

As the `jolux:Act` is a very abstract representation of a legislative resource, there is e.g. no title of the law attached to the `jolux:Act` because this is something language specific and therefore added to the `jolux:Expression` of the `jolux:Act`.

As the Official Compilation is released in a weekly bulletin, all `jolux:Act` are part of such a bulletin via

`jolux:isPartOf` and the bulletin itself is a type `jolux:Memorial`:

`jolux:isPartOf`

The object property `jolux:isPartOf` is used to connect a `jolux:Act` to the weekly bulletin that it is part of.

`jolux:Memorial`

The owl:Class `jolux:Memorial` is used for the weekly bulletin that contains the new entries of the Official Compilation and for the Federal Gazette.

3.4 Datatype Properties

- `jolux:publicationDate`
- `jolux:dateEndApplicability`
- `jolux:dateEntryInForce`
- `jolux:dateNoLongerInForce`
- `jolux:dateDocument`

3.5 Object Properties

Object properties that point to a vocabulary entry:

- `jolux:processType`
- `jolux:typeDocument`
- `jolux:classifiedByTaxonomyEntry`
- `jolux:legalResourceGenre`
- `jolux:responsibilityOf`

Object properties that point to an individual:

- `jolux:isRealizedBy`
- `jolux:isPartOf`

3.6 SPARQL Examples

The following SPARQL query shows all the different `jolux:Expression` for the federal constitution:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?expression WHERE {
  <https://fedlex.data.admin.ch/eli/oc/1999/404> jolux:isRealizedBy ?expression.
}
```

The following SPARQL query shows all the different `jolux:Manifestation` for the federal constitution:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?manifestation WHERE {
  <https://fedlex.data.admin.ch/eli/oc/1999/404> jolux:isRealizedBy ?expression.
```

```
}  
    ?expression jolux:isEmbodiedBy ?manifestation.  
}
```

The following SPARQL query shows all the different jolux:Act that have the legal genre “Basic legislation” and are not yet in force.

```
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>  
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>  
SELECT ?act ?date WHERE {  
    ?act jolux:legalResourceGenre <https://fedlex.data.admin.ch/vocabulary/  
/legal-resource-genre/100>;  
    jolux:dateEntryInForce ?date.  
    FILTER(?date > xsd:date(NOW()))  
}
```

Classified Compilation (CC)

The *Classified Compilation* (CC) (also known as Systematic Compilation) is according to the [lexicon of parliamentary terms](#) a regularly updated and revised collection of the legislative resources of the Official Compilation arranged under subject headings.

An important aspect of the Classified Compilation is the classification of the entries according to a legal taxonomy. A deeper explanation of the Classified Compilation is also available in [German](#) /[French](#)/[Italian](#).

This part explains all the important objects that build an entry in the Classified Compilation and it does so with the help of the federal constitution as an example of an entry in the Classified Compilation.

Hint for legal laypersons

Entries in the Classified Compilation are consolidations of entries in the Official Compilation. The main reason for having a Classified Compilation is a better usability of the legislative resources because the Classified Compilation represents the current state of a legislative resource.

It is important to realize that the Classified Compilation is not legally binding, the source of the “true law” is always the Official Compilation.

4.1 Example

Throughout this sub-page, the federal constitution is used as an example of an entry in the Classified Compilation.

- URI: <https://fedlex.data.admin.ch/eli/cc/1999/404>
- URL: <https://www.fedlex.admin.ch/eli/cc/1999/404>
- [Metadata viewer](#)

4.2 URI

The URI of an entry in the Classified Compilation contains the following parts:

- Standard namespace and path: `https://fedlex.data.admin.ch/eli/`
- the part `cc/` denotes the Classified Compilation, meaning that this URI identifies something that is part of the Classified Compilation of the federal law
- `YYYY/` is the year of the publication
- `ID` an identifier that has no specific meaning

The ID part has no specific meaning but is following the basic act in the Official Compilation. So the URI of an entry in the Classified Compilation can be created by replacing the `oc` in the URI of the basic act with `cc`. E.g. the URI of the federal constitution in the Official Compilation is `https://fedlex.data.admin.ch/eli/oc/1999/404` and the URI of the federal constitution in the Classified Compilation is `https://fedlex.data.admin.ch/eli/cc/1999/404`. This does not hold true for cases where the base act is not part of the Official Compilation but of the Federal Gazette. In this case, the `fga` part is replaced by `cc` and a suffix `_fga` is appended to the ID part E.g. `https://fedlex.data.admin.ch/eli/fga/2012/1262` becomes `https://fedlex.data.admin.ch/eli/cc/2012/1262_fga`. There are other special cases like `https://fedlex.data.admin.ch/eli/cc/2012/136` is based on `https://fedlex.data.admin.ch/eli/fga/2008/1057`.

4.3 General Structure

Every entry in the Classified Compilation is of type `jolux:ConsolidationAbstract`.

`jolux:ConsolidationAbstract`

The owl:Class `jolux:ConsolidationAbstract` is used for entries in the Classified Compilation.

It is a consolidation because it consolidates different entries from the Official Compilation into a single document that shows the current state. The term *abstract* is not so much meant as a summary but as an abstraction.

A `jolux:ConsolidationAbstract` has a `jolux:Expression` attached for representing the title and abbreviation in different languages of this consolidation because this does not change. But there are no `jolux:Manifestation` these only exist for `jolux:Consolidation`.

`jolux:Consolidation`

The owl:Class `jolux:Consolidation` is used for versions that represent a `jolux:ConsolidationAbstract` at a specific time. It is of the same [abstraction level](#) as `jolux:Work` and all `jolux:Consolidation` are also `jolux:Work`.

The different `jolux:Consolidation` are no “deltas” of the changes but always the complete state at the specific point in time.

For `jolux:Consolidation`, the additional [abstraction levels](#) `jolux:Expression` and `jolux:Manifestation` are usually also available for all entries. This excludes some older ones and those that are not yet published.

The connection between `jolux:Consolidation` and `jolux:ConsolidationAbstract` is made with `jolux:isMemberOf`.

`jolux:isMemberOf`

The object property `jolux:isMemberOf` is used to connect a [jolux:Consolidation](#) to a [jolux:ConsolidationAbstract](#). It is also used to connect the weekly bulletins of the Official Compilation to the yearly collection.

Each `jolux:ConsolidationAbstract` is based on an `jolux:Act` through `jolux:basicAct`.

`jolux:basicAct`

The object property `jolux:basicAct` is used to connect a [jolux:ConsolidationAbstract](#) to a `jolux:Act`. The connected act is the first version of the consolidation.

The following figure shows the general structure of an entry in the Classified Compilation:




Figure 4.1. General structure of an entry in the Classified Compilation.

4.4 Legal Taxonomy

As the name Classified Compilation suggests, there is a classification scheme that sorts the entries in the Classified Compilation according to a legal taxonomy. This taxonomy is hierarchical and formulated as [vocabulary](#).

4.5 Datatype Properties

4.5.1 jolux:ConsolidationAbstract

- [jolux:dateEntryInForce](#)
- [jolux:dateDocument](#)
- [jolux:dateNoLongerInForce](#)
- [jolux:dateEndApplicability](#)

4.5.2 jolux:Consolidation

- jolux:publicationDate
- jolux:dateApplicability
- jolux:dateEndApplicability

4.6 Object Properties

4.6.1 jolux:ConsolidationAbstract

Object properties that point to a vocabulary entry:

- jolux:typeDocument
- jolux:classifiedByTaxonomyEntry
- jolux:inForceStatus

Object properties that point to an individual:

- jolux:basicAct
- jolux:isRealizedBy

4.6.2 jolux:Consolidation

Object properties that point to an individual:

- jolux:isMemberOf
- jolux:isRealizedBy

4.7 SPARQL Examples

The following SPARQL query shows all the different versions of the federal constitution:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {
    ?consolidation jolux:isMemberOf <https://fedlex.data.admin.ch/eli/cc/1999/404>.
}
```

The following SPARQL query gives the PDF link to the latest version of the constitution in English through a chain to jolux:Consolidation, jolux:Expression and jolux:Manifestation:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {
    ?work jolux:isMemberOf <https://fedlex.data.admin.ch/eli/cc/1999/404>;
        jolux:dateApplicability ?date;
        jolux:isRealizedBy ?expression.
    ?expression jolux:language <http://publications.europa.eu/resource/authority/language/ENG>;
        jolux:isEmbodiedBy ?manifestation.
    ?manifestation jolux:format <http://publications.europa.eu/resource/authority/file-type/PDF>;
        jolux:isExemplifiedBy ?url.
} ORDER BY DESC(?date)
```



```
LIMIT 1
```

Impacts

Impacts is what connect entries in the Classified Compilation and Official Compilation. The entries in the Classified Compilation are consolidations of entries in the **Official Compilation**. This means that entries in the Official Compilation usually have an impact on entries in the Classified Compilation.

Impacts are modelled as `jolux:LegalResourceImpact`.

`jolux:LegalResourceImpact`

The owl:Class `jolux:LegalResourceImpact` is used to build an entry in the Classified Compilation out of different entries in the Official Compilation. Entries in the Official Compilation have impacts on entries in the Classified Compilation. The `jolux:LegalResourceImpact` has two main predicates. `jolux:impactFromLegalResource` points to the source of the impact and `jolux:impactToLegalResource` points to the impacted resource.

5.1 Example

Throughout this sub-page, the following `jolux:LegalResourceImpact` is used as an example.

- URI: <https://fedlex.data.admin.ch/eli/oc/2015/104/legal-analysis/LegalResourceImpact/1>
- URL: No URL available for `jolux:LegalResourceImpact`
- [Metadata viewer](#)

5.2 URI

The URI of a `jolux:LegalResourceImpact` contains the following parts:

- it starts with the URI of the entry in the Official Compilation that is the source for the impact.
- `/legal-analysis/LegalResourceImpact/` denotes all impacts
- ID an identifier that has no specific meaning

5.3 General Structure

The following figure shows the connection between entries in the Official Compilation and Classified Compilation through `jolux:LegalResourceImpact`:




Figure 5.1. Connection between entries in the Official Compilation and the Classified Compilation through impacts.

As it is shown in the figure above, the connection between `jolux:LegalResourceImpact` and the entries in the Official Compilation and Classified Compilation is not direct but through `jolux:LegalResourceSubdivision`. The predicates used to connect the source and the target of the impact are `jolux:impactFromLegalResource` and `jolux:impactToLegalResource`.

`jolux:impactFromLegalResource`

The object predicate **`jolux:impactFromLegalResource`** is used to connect a `jolux:LegalResourceImpact` to a `jolux:LegalResourceSubdivision` as a source of the impact.

`jolux:impactToLegalResource`

The object predicate **`jolux:impactToLegalResource`** is used to connect a `jolux:LegalResourceImpact` to a `jolux:LegalResourceSubdivision` as a target of the impact.

`jolux:impactConsolidatedBy`

The object predicate **`jolux:impactConsolidatedBy`** is used to connect a `jolux:LegalResourceImpact` to a `jolux:LegalResourceSubdivision` that is part of the corresponding `jolux:Consolidation`.

5.4 Datatype Properties

- `jolux:legalResourceImpactHasDateEntryInForce`

5.5 Object Properties

Object properties that point to a vocabulary entry:

- `jolux:legalResourceImpactHasType`

Object properties that point to an individual:

- `jolux:impactFromLegalResource`
- `jolux:impactToLegalResource`
- `jolux:impactConsolidatedBy`

5.6 SPARQL Examples

The following query shows all the entries in the Official Compilation, that have an impact on the federal constitution:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT DISTINCT ?act ?date WHERE {
  ?impact jolux:impactFromLegalResource/jolux:legalResourceSubdivisionIsPartOf
  ?act;
         jolux:impactToLegalResource/jolux:legalResourceSubdivisionIsPartOf <https://fedlex.data.admin.ch/eli/cc/1999/404>.
  ?act jolux:dateEntryInForce ?date.
} ORDER BY ?date
```

Subdivisions

To structure a legislative resource further, `jolux:LegalResourceSubdivision` is used:

`jolux:LegalResourceSubdivision`

The class `jolux:LegalResourceSubdivision` is used to structure each legislative resource into units: Article (basic unit) and elements above and below this in the hierarchy, as well as annexes and other elements. The concrete unit is attached by using `jolux:legalResourceSubdivisionType`.

6.1 Example

Throughout this sub-page, the following `jolux:LegalResourceSubdivision` is used as an example.

- URI: <https://fedlex.data.admin.ch/eli/cc/1999/404/text>
- URL: No URL available for `jolux:LegalResourceSubdivision`
- [Metadata viewer](#)

6.2 URI

The URI of a `jolux:LegalResourceSubdivision` contains the following parts:

- it starts with the URI of the entry in the Official or Consolidated Compilation
- `/type` denotes the type of the subdivision

6.3 General Structure

The following figure shows the general structure of a `jolux:LegalResourceSubdivision`:




Figure 6.1. General structure of a jolux:LegalResourceSubdivision.

The jolux:LegalResourceSubdivision is bound to a jolux:Work through jolux:legalResourceSubdivisionIsPartOf.

jolux:legalResourceSubdivisionIsPartOf

The property **jolux:legalResourceSubdivisionIsPartOf** is used to connect a jolux:LegalResourceSubdivision with its [jolux:Work](#).

The jolux:LegalResourceSubdivision have a type from the [subdivision types vocabulary](#).

Hint for legal laypersons

With help of jolux:LegalResourceSubdivision, it is possible to divide a legislative resource also on the level of the individual [articles](#). But this is not systematically done (yet). The single articles are only modelled when it is necessary for a [jolux:LegalResourceImpact](#) (see [Section 6.5](#) below).

6.4 Object Properties

Object properties that point to a vocabulary entry:

- [jolux:legalResourceSubdivisionType](#)

Object properties that point to an individual:

- [jolux:legalResourceSubdivisionIsPartOf](#)

6.5 SPARQL Examples

The following query shows all the subdivisions of the federal constitution in the Classified Compilation with its types:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {
  ?subdivison jolux:legalResourceSubdivisionIsPartOf <https://fedlex.data.admin.ch/eli/cc/1999/404>;
  jolux:legalResourceSubdivisionType ?type.
}
```

History

The term **history** is used to list all legislative resources that are relevant for a specific entry in the [Classified Compilation](#).

There is the possibility to see all relevant events for a specific entry in the [Classified Compilation](#) in the web frontend by adding `/history` to the URL of the entry. E.g. for seeing all the events for the Federal Constitution (not available in English), the URL would be `https://www.fedlex.admin.ch/eli/cc/1999/404/de/history`.

There are different mechanisms how legislative resources can be relevant for the history:

7.1 Draft

In creating a new legislative resource, a [jolux:Draft](#) is created.

jolux:Draft

A **jolux:Draft** is used to bundle all the activities and documents during the process of drafting a new legislative resource.

The following figure shows the general structure of a [jolux:Draft](#):




Figure 7.1. General structure of a jolux:Draft.

The following SPARQL query shows all the `jolux:Act` that were involved in drafting the Federal Constitution:

```

PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?act WHERE {
  ?draft jolux:hasResultingLegalResource <https://fedlex.data.admin.ch/eli/oc/1999/404>;
  jolux:draftHasLegislativeTask ?task.
  ?task jolux:legislativeTaskHasResultingLegalResource ?act.
}

```

7.2 Impact

If new legislative resources are created that have an impact on existing ones, they are modelled as Figure 5.1.

The following SPARQL query shows all the `jolux:Act` that have an impact on the Federal Constitution:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT DISTINCT ?act WHERE {
    ?impact jolux:impactFromLegalResource/jolux:legalResourceSubdivisionIsPartOf
    ?act;
        jolux:impactToLegalResource/jolux:legalResourceSubdivisionIsPartOf <https://fedlex.data.admin.ch/eli/cc/1999/404>.
}
```

Changes

The term **changes** is used to list all changes to a specific entry in the [Classified Compilation](#) on the level of single articles.

There is the possibility to see all changes of a specific entry in the [Classified Compilation](#) in the web frontend by adding `/changes` to the URL of the entry. E.g. for seeing all the changes of the Federal Constitution (not available in English), the URL would be `https://www.fedlex.admin.ch/eli/cc/1999/404/de/changes`.

Changes are extracted from [Figure 5.1](#). There are two methods used to depict the concerned articles depending on the date of the change.

8.1 Older Changes

The concerned articles of older changes are modelled via a `jolux:impactToLegalResourceComment`:

`jolux:impactToLegalResourceComment`

The datatype property `jolux:impactToLegalResourceComment` lists the impacted articles on the legislative resource from the [Figure 5.1](#) as `rdf:langString`.

8.2 Newer Changes

For newer changes, the impacted articles are modelled as `jolux:LegalResourceSubdivision`.

8.3 SPARQL Examples

The following sparql query lists all the impacts on the Federal Constitution. The concerned articles are either given in the `jolux:impactToLegalResourceComment` or modelled as `jolux:LegalResourceSubdivision`:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {

    ?subdivision jolux:legalResourceSubdivisionIsPartOf <https://fedlex.data.admin.ch/eli/cc/1999/404>.
    ?impact jolux:impactToLegalResource ?subdivision;
        jolux:impactFromLegalResource/jolux:legalResourceSubdivisionIsPartOf ?act;
        jolux:legalResourceImpactHasType ?type.

    OPTIONAL {
```

```
        ?impact jolux:impactToLegalResourceComment ?comment.  
    }  
  
    FILTER (!BOUND(?comment) || lang(?comment) = "de")  
}
```

Citations

Legal texts often refer other legislative resources in the form of citations. In JOLux, citations are modelled as `jolux:Citation`.

jolux:Citation

A **jolux:Citation** is used to model citations between legislative resources.

The following figure shows the general structure of a citation:




Figure 9.1. General structure of a citation.

The citations for an entry in the [Classified Compilation](#) can be shown with the following URL (example for the Federal Constitution): <https://www.fedlex.admin.ch/eli/cc/1999/404/de/quotes>. These pages are only available in German, French, and Italian.

9.1 From and To

A citation *from* means the citing resource whereas *to* means the cited resource. So if a legislative resource cites the Federal Constitution, the citation goes *from* this resource *to* the Federal Constitution.

jolux:citationFromLegalResource

The object property **jolux:citationFromLegalResource** is used to connect a **jolux:Citation** to a **jolux:LegalResourceSubdivision** that is the **citing** document.

jolux:citationToLegalResource

The object property **jolux:citationToLegalResource** is used to connect a **jolux:Citation** to a **jolux:LegalResourceSubdivision** that is the **cited** document.

9.2 SPARQL Example

The following SPARQL query shows all the **jolux:ConsolidationAbstract** with its German titles that cite the Federal Constitution:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {
    ?citation jolux:citationToLegalResource/jolux:legalResourceSubdivisionIsPartOf
    <https://fedlex.data.admin.ch/eli/cc/1999/404>;
                                jolux:citationFromLegalResource
    /jolux:legalResourceSubdivisionIsPartOf/jolux:isMemberOf
    ?consolidationAbstractFrom.
    ?citation jolux:descriptionFrom ?descriptionFrom;
            jolux:language <http://publications.europa.eu/resource/authority
    /language/DEU>.
    ?consolidationAbstractFrom jolux:isRealizedBy ?expressionFrom.
    ?expressionFrom jolux:language <http://publications.europa.eu/resource/authority
    /language/DEU>;
            jolux:title ?titleFrom.
}
```

International Treaties

According to [Termdat](#), an international treaty is an “international agreement concluded between states in written form and governed by international law, whether embodied in a single instrument or in two or more related instruments and whatever its particular designation”.

Hint for legal laypersons

International Treaties are sometimes but not always approbated into the [Official Compilation](#).

10.1 Example

Throughout this sub-page, the following treaty process is used as an example.

- URI: <https://fedlex.data.admin.ch/eli/treaty/2024/0311>
- URL: <https://www.fedlex.admin.ch/eli/treaty/2024/0311/de>
- [Metadata viewer](#)

10.2 URI

The URI of a treaty process contains the following parts:

- Standard namespace and path: `https://fedlex.data.admin.ch/eli/`
- the part `treaty/` shows that it is about a treaty
- `YYYY/` is the year of the publication
- `ID` an identifier that has no specific meaning

10.3 General Structure

Treaties evolve around a `jolux:TreatyProcess`.

`jolux:TreatyProcess`

The owl:Class `jolux:TreatyProcess` is used group the important elements of an international treaty.

The following figure shows the general structure of a treaty process:




Figure 10.1. General structure of a treaty process.

A `jolux:TreatyProcess` is described by the following properties:

jolux:bilateral

The datatype property **jolux:bilateral** is used to classify whether a treaty is bilateral. The datatype is `xsd:boolean`.

jolux:titleTreaty

The datatype property **jolux:titleTreaty** links to the title of the treaty. The datatype is `rdf:langString`.

jolux:treatySignatureDate

The datatype property **jolux:treatySignatureDate** links to the date of the signature of the treaty. The datatype is `xsd:date`.

jolux:treatySignaturePlace

The datatype property **jolux:treatySignaturePlace** links to the place of the signature of the treaty. The datatype is `xsd:string`.

Entities of class `jolux:TreatyProcess` have different `jolux:TaskForTreaty` added.

jolux:TaskForTreaty

The owl:Class **jolux:TaskForTreaty** is used as class for all the tasks that are necessary for concluding a treaty.

The tasks are linked to the process via `jolux:treatyProcessHasTask`.

jolux:treatyProcessHasTask

The object property **jolux:treatyProcessHasTask** links a `jolux:TreatyProcess` to a `jolux:TaskForTreaty`.

The possible types of tasks are given in the section [Section 10.6](#).

The result of an international treaty is always a `jolux:TreatyDocument`.

jolux:TreatyDocument

The owl:Class **jolux:TreatyDocument** is used for the resulting treaty of the [jolux:TreatyProcess](#).

The `jolux:TreatyDocument` is linked via `jolux:treatyProcessHasResultingTreatyDocument`.

jolux:treatyProcessHasResultingTreatyDocument

The object property **jolux:treatyProcessHasResultingTreatyDocument** links a `jolux:TreatyProcess` to a `jolux:TreatyDocument`.

If there is an approbation into the [Official Compilation](#) The approbation act is linked to the treaty process via `jolux:approbationAct`.

jolux:approbationAct

The object property **jolux:approbationAct** links a `jolux:TreatyProcess` to a `jolux:Act` in the [Official Compilation](#).

10.4 Datatype Properties for jolux:TreatyProcess

- jolux:bilateral
- jolux:titleTreaty
- jolux:treatySignatureDate
- jolux:treatySignaturePlace

10.5 Object Properties for jolux:TreatyProcess

Object properties that point to a vocabulary entry:

- jolux:treatySubject
- jolux:treatyType
- jolux:treatyPartyCountry
- jolux:classifiedByTaxonomyEntry

Object properties that point to an individual:

- jolux:treatyProcessHasTask
- jolux:treatyProcessHasResultingTreatyDocument
- jolux:approbationAct

10.6 SPARQL Examples

The following SPARQL query shows all the different classes that are used on a jolux:TaskForTreaty to further segment these tasks:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT DISTINCT ?class WHERE {
  ?task a jolux:TaskForTreaty;
        a ?class.
  FILTER(?class != <http://data.legilux.public.lu/resource/ontology/jolux#Event>)
}
```

The class jolux:Event is filtered because all jolux:TaskForTreaty are also jolux:Event.

The following SPARQL query shows the 100 newest treaties that have an English title with the approbation act if available:

```
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT * WHERE {

  ?treaty_process a jolux:TreatyProcess;
                 jolux:treatySignatureDate ?date;
                 jolux:titleTreaty ?title.

  OPTIONAL {
    ?treaty_process jolux:approbationAct ?act.
  }
}
```



```
FILTER(lang(?title) = "en")  
FILTER(?title != ""@en)  
  
}  
ORDER BY DESC(?date)  
LIMIT 100
```


Vocabularies

In RDF, a vocabulary is a **set of predefined terms**, that is used to describe resources. These terms are defined in a way that allows them to be used **consistently across different datasets**. In the case of JOLux, vocabularies are specifically used only on the object position of RDF triples.

Fedlex defines and makes use of multiple vocabularies. This sub-page lists an overview and the main vocabularies and its associated properties that are used in describing the legislative resources.

11.1 Available Vocabularies

As all vocabularies are modelled as having the class `skos:ConceptScheme`, the metadata viewer can give all the vocabularies as incoming relations to `skos:ConceptScheme` and therefore serves as an **overview on all available vocabularies**.

In addition, the following SPARQL query shows all vocabularies with its name in English, German and French. The name is either `dcterms:title` where available and otherwise `rdfs:label` or empty if there is neither:

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
SELECT DISTINCT ?vocabulary ?name_en ?name_de ?name_fr WHERE {
    ?vocabulary a skos:ConceptScheme.

    # English title
    OPTIONAL {
        ?vocabulary dcterms:title ?title_en .
        FILTER(LANG(?title_en) = "en")
    }

    # German title
    OPTIONAL {
        ?vocabulary dcterms:title ?title_de .
        FILTER(LANG(?title_de) = "de")
    }

    # French title
    OPTIONAL {
        ?vocabulary dcterms:title ?title_fr .
        FILTER(LANG(?title_fr) = "fr")
    }

    # English label
    OPTIONAL {
        ?vocabulary rdfs:label ?label_en .
        FILTER(LANG(?label_en) = "en")
    }
}
```

```
}

# German label
OPTIONAL {
  ?vocabulary rdfs:label ?label_de .
  FILTER(LANG(?label_de) = "de")
}

# French label
OPTIONAL {
  ?vocabulary rdfs:label ?label_fr .
  FILTER(LANG(?label_fr) = "fr")
}

# Use ?title if available and otherwise ?label
BIND(COALESCE(?title_en, ?label_en) AS ?name_en)
BIND(COALESCE(?title_de, ?label_de) AS ?name_de)
BIND(COALESCE(?title_fr, ?label_fr) AS ?name_fr)
}
```

11.2 Hierarchical Vocabularies

Some vocabularies are modelled as hierarchy or taxonomy of entries. The following SPARQL query lists all vocabularies that use a hierarchy:

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
SELECT DISTINCT ?vocabulary ?name_en WHERE {
  ?vocabulary a skos:ConceptScheme;
  skos:hasTopConcept ?top_concept.
  ?top_concept skos:narrower ?narrow.

  # English title
  OPTIONAL {
    ?vocabulary dcterms:title ?title_en .
    FILTER(LANG(?title_en) = "en")
  }

  # English label
  OPTIONAL {
    ?vocabulary rdfs:label ?label_en .
    FILTER(LANG(?label_en) = "en")
  }

  # Use ?title if available and otherwise ?label
  BIND(COALESCE(?title_en, ?label_en) AS ?name_en)
}
```

11.3 Act Types

Act Types

- URI: <https://fedlex.data.admin.ch/vocabulary/legal-resource-genre>
- Description: The **act types** vocabulary is used to classify the type of a jolux:Act.
- Predicates: jolux:legalResourceGenre

- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary
/legal-resource-genre>;
    skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "en")
}
```

11.4 Countries

Countries

- URI: <https://fedlex.data.admin.ch/vocabulary/country>
- Description: The **countries** vocabulary is used to link to a specific country.
- Predicates: jolux:treatyPartyCountry
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/country>;
    skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "en")
}
```

11.5 Enforcement Status

Enforcement Status

- URI: <https://fedlex.data.admin.ch/vocabulary/enforcement-status>
- Description: The **enforcement status** vocabulary is used to classify the type of a jolux:ConsolidationAbstract.
- Predicates: jolux:inForceStatus
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary
```

```
/enforcement-status>;
    skos:prefLabel ?label.
FILTER NOT EXISTS { ?term a skos:Collection }
FILTER (lang(?label) = "en")
}
```

11.6 Impact Types

Impact Types

- URI: <https://fedlex.data.admin.ch/vocabulary/impact-type>
 - Description: The **impact types** vocabulary is used to classify the type of a jolux:LegalResourceImpact.
 - Predicates: jolux:legalResourceImpactHasType
 - [Metadata viewer](#)
-

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/impact-type>;
    skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "en")
}
```

11.7 Legal Institution

Legal Institution

- URI: <https://fedlex.data.admin.ch/vocabulary/legal-institution>
 - Description: The **legal institution** vocabulary is used to add the responsibility of an institution for jolux:Act.
 - Predicates: jolux:responsibilityOf
 - [Metadata viewer](#)
-

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/
/legal-institution>;
    skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "en")
}
```

As this is a hierarchical vocabulary, the following SPARQL query shows the hierarchy of the entries:

```

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

SELECT (GROUP_CONCAT(CONCAT(STR(?endpoint_level), ": ", STR(?endpoint_label));
separator = " <---- ") AS ?hierarchy) WHERE {
  SELECT * WHERE {
    ?endpoint skos:narrower* ?intermediate;
    skos:prefLabel ?endpoint_label.
    FILTER(lang(?endpoint_label) = "de")
  }
  {
    SELECT ?endpoint (COUNT(?endpoint) as ?endpoint_level) WHERE {
      BIND (<https://fedlex.data.admin.ch/vocabulary/legal-institution>
as ?root)
      ?root skos:hasTopConcept/skos:narrower* ?intermediate.
      ?intermediate skos:narrower* ?endpoint.
    } GROUP BY ?endpoint ORDER BY ?endpoint_level
  }
  ORDER BY ?intermediate ?endpoint_level
} GROUP BY ?intermediate ORDER BY ?hierarchy

```

11.8 Legal Taxonomy

Legal Taxonomy

- URI: <https://fedlex.data.admin.ch/vocabulary/legal-taxonomy>
- Description: The **legal taxonomy** vocabulary is used to classify entries of a `jolux:ConsolidationAbstract`.
- Predicates: `jolux:classifiedByTaxonomyEntry`
- Metadata viewer

The following SPARQL query shows all the entries of this vocabulary with its labels:

```

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
  ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/legal-taxonomy>;
  skos:prefLabel ?label.
  FILTER NOT EXISTS { ?term a skos:Collection }
  FILTER (lang(?label) = "en")
}

```

As this is a hierarchical vocabulary, the following SPARQL query shows the hierarchy of the entries:

```

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

SELECT (GROUP_CONCAT(CONCAT(STR(?endpoint_level), ": ", STR(?endpoint_label));
separator = " <---- ") AS ?hierarchy) WHERE {
  SELECT * WHERE {
    ?endpoint skos:narrower* ?intermediate;
    skos:prefLabel ?endpoint_label.
    FILTER(lang(?endpoint_label) = "de")
  }
  {
    SELECT ?endpoint (COUNT(?endpoint) as ?endpoint_level) WHERE {
      BIND (<https://fedlex.data.admin.ch/vocabulary/legal-taxonomy> as
?root)
      ?root skos:hasTopConcept/skos:narrower* ?intermediate.
      ?intermediate skos:narrower* ?endpoint.
    }
  }
  ORDER BY ?intermediate ?endpoint_level
} GROUP BY ?intermediate ORDER BY ?hierarchy

```

```
        } GROUP BY ?endpoint ORDER BY ?endpoint_level
    }
    } ORDER BY ?intermediate ?endpoint_level
} GROUP BY ?intermediate ORDER BY ?hierarchy
```

As this is a very important vocabulary, there are also two special webpages that show these hierarchical legal taxonomy entries. One for [Swiss law](#) and one for [international law](#).

11.9 Procedure Types

Procedure Types

- URI: <https://fedlex.data.admin.ch/vocabulary/type-procedure>
- Description: The **procedure types** vocabulary is used to classify the type of a jolux:Act.
- Predicates: jolux:processType
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/type-procedure>;
    skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "en")
}
```

11.10 Subdivision Types

Subdivision Types

- URI: <https://fedlex.data.admin.ch/vocabulary/subdivision-type>
- Description: The **subdivision types** vocabulary is used to classify the type of a jolux:LegalResourceSubdivision.
- Predicates: jolux:legalResourceSubdivisionType
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels:

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/subdivision-type>;
    skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "en")
}
```

As this is a hierarchical vocabulary, the following SPARQL query shows the hierarchy of the entries:


```

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

SELECT (GROUP_CONCAT(CONCAT(STR(?endpoint_level), ": ", STR(?endpoint_label));
separator = " <---- ") AS ?hierarchy) WHERE {
  SELECT * WHERE {
    ?endpoint skos:narrower* ?intermediate;
    skos:prefLabel ?endpoint_label.
    FILTER(lang(?endpoint_label) = "de")
  }
  {
    SELECT ?endpoint (COUNT(?endpoint) as ?endpoint_level) WHERE {
      BIND (<https://fedlex.data.admin.ch/vocabulary/consultation-stage>
as ?root)
      ?root skos:hasTopConcept/skos:narrower* ?intermediate.
      ?intermediate skos:narrower* ?endpoint.
    } GROUP BY ?endpoint ORDER BY ?endpoint_level
  }
} ORDER BY ?intermediate ?endpoint_level
} GROUP BY ?intermediate ORDER BY ?hierarchy

```

11.11 Text Types

Text Types

- URI: <https://fedlex.data.admin.ch/vocabulary/resource-type>
- Description: The **text types** vocabulary is used to classify the text type of a jolux:Work.
- Predicates: jolux:typeDocument, jolux:historicalTypeDocument
- Metadata viewer

The following SPARQL query shows all the entries of this vocabulary with its labels:

```

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
  ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/resource-type>;
  skos:prefLabel ?label.
  FILTER NOT EXISTS { ?term a skos:Collection }
  FILTER (lang(?label) = "en")
}

```

As this is a hierarchical vocabulary, the following SPARQL query shows the hierarchy of the entries:

```

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

SELECT (GROUP_CONCAT(CONCAT(STR(?endpoint_level), ": ", STR(?endpoint_label));
separator = " <---- ") AS ?hierarchy) WHERE {
  SELECT * WHERE {
    ?endpoint skos:narrower* ?intermediate;
    skos:prefLabel ?endpoint_label.
    FILTER(lang(?endpoint_label) = "de")
  }
  {
    SELECT ?endpoint (COUNT(?endpoint) as ?endpoint_level) WHERE {
      BIND (<https://fedlex.data.admin.ch/vocabulary/resource-type> as
?root)
      ?root skos:hasTopConcept/skos:narrower* ?intermediate.
      ?intermediate skos:narrower* ?endpoint.
    } GROUP BY ?endpoint ORDER BY ?endpoint_level
  }
}

```

```
    }  
  } ORDER BY ?intermediate ?endpoint_level  
} GROUP BY ?intermediate ORDER BY ?hierarchy
```

11.12 Treaty Subject Themes

Treaty Subject Themes

- URI: <https://fedlex.data.admin.ch/vocabulary/treaty-subject-theme>
- Description: The **treaty subject themes** vocabulary is used to classify the subject of a `jolux:TreatyProcess`.
- Predicates: `jolux:treatySubject`
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels (only in German because there are no English labels):

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>  
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>  
SELECT ?term ?label WHERE {  
  ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/  
/treaty-subject-theme>;  
  skos:prefLabel ?label.  
  FILTER NOT EXISTS {?term a skos:Collection}  
  FILTER (lang(?label) = "de")  
}
```

11.13 Treaty Types

Treaty Types

- URI: <https://fedlex.data.admin.ch/vocabulary/treaty-type>
- Description: The **treaty types** vocabulary is used to classify the type of a `jolux:TreatyProcess`.
- Predicates: `jolux:treatyType`
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels (only in German because there are no English labels):

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>  
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>  
SELECT ?term ?label WHERE {  
  ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/treaty-type>;  
  skos:prefLabel ?label.  
  FILTER NOT EXISTS {?term a skos:Collection}  
  FILTER (lang(?label) = "de")  
}
```

11.14 User Formats

User Formats

- URI: <https://fedlex.data.admin.ch/vocabulary/user-format>
- Description: The **user formats** vocabulary is used to classify the file type of a `jolux:Manifestation`.
- Predicates: `jolux:userFormat`
- [Metadata viewer](#)

The following SPARQL query shows all the entries of this vocabulary with its labels (only in German because there are no English labels):

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX jolux: <http://data.legilux.public.lu/resource/ontology/jolux#>
SELECT ?term ?label WHERE {
    ?term skos:inScheme <https://fedlex.data.admin.ch/vocabulary/user-format>;
        skos:prefLabel ?label.
    FILTER NOT EXISTS { ?term a skos:Collection }
    FILTER (lang(?label) = "de")
}
```

Dates

Concerning legislation, there are a number of dates that are important during the legislative process. All the date literals are modelled with datatype `xsd:date`. The following list tries to give the most important date properties according the lifetime of a legal resource.

jolux:dateDocument

The property `jolux:dateDocument` is used to indicate the decision date of an act, or the date on which the document was adopted. For bills, where there is no formal adoption date, the Document Date is used as the reference date.

jolux:publicationDate

The property `jolux:publicationDate` denotes the date of first publication in one of the collections. If the same document is republished, this date does not change: it is always the date of the initial publication that is decisive. On the other hand, for consolidations, several publication dates may exist: each new publication of the file generates a new date.

Hint for legal laypersons

During introduction of a new law, there are two similar but different dates relevant: the date of the **entry into force** and the date of the **applicability** of the law. These two dates can differ. It can happen that the entry into force comes before the law is applicable. This is mainly to allow for preparing the necessary changes in the affected organizations. During this period of time there could be some transitional provisions in place.

jolux:dateEntryInForce

The property `jolux:dateEntryInForce` denotes the date on which the law starts to have a legal effect (see above).

jolux:dateApplicability

The property `jolux:dateApplicability` denotes the date from which a law is applicable (see above).

Hint for legal laypersons

During cancellation of a law, there are two similar but different dates relevant: the date of the **end of applicability** and the date of the **no longer in force**. The end of applicability is the last day where the law is applicable. The no longer in force date is the first day that the law has no more a legal effect. These two dates can differ in both direction. It could be that the law is first no more applicable and no

longer in force later or vice versa.

jolux:dateEndApplicability

The property `jolux:dateEndApplicability` denotes the last day where a law is applicable.

jolux:dateNoLongerInForce

The property `jolux:dateNoLongerInForce` denotes the first day on which the law has no more a legal effect.

12.1 Special Dates

jolux:legalResourceImpactHasDateEntryInForce

The property `jolux:legalResourceImpactHasDateEntryInForce` denotes the effective date of an [`jolux:LegalResourceImpact`](#).

Reference

The following terms are all to be understood as prefixed with `jolux:`, so the full URI for e.g. `Act` is: `http://data.legilux.public.lu/resource/ontology/jolux#Act`. The terms are sorted alphabetically.

13.1 JOLux Classes

- `Act`
- `Citation`
- `Consolidation`
- `ConsolidationAbstract`
- `Draft`
- `Expression`
- `LegalResourceImpact`
- `LegalResourceSubdivision`
- `Manifestation`
- `TaskForTreaty`
- `TreatyDocument`
- `TreatyProcess`
- `Work`

13.2 JOLux Datatype Properties

Dates:

- `dateApplicability`
- `dateDocument`
- `dateEndApplicability`
- `dateEntryInForce`
- `dateNoLongerInForce`
- `legalResourceImpactHasDateEntryInForce`
- `publicationDate`

Others:

- bilateral
- impactToLegalResourceComment
- treatySignatureDate
- treatySignaturePlace
- titleTreaty

13.3 JOLux Object Properties

Link to vocabulary entries:

- classifiedByTaxonomyEntry
- historicalTypeDocument
- inForceStatus
- legalRessourceGenre
- legalResourceImpactHasType
- legalResourceSubdivisionType
- processType
- responsibilityOf
- treatyPartyCountry
- treatySubject
- treatyType
- typeDocument
- userFormat

Link to individuals:

- approbationAct
- basicAct
- citationFromLegalResource
- citationToLegalResource
- impactConsolidatedBy
- impactFromLegalResource
- impactToLegalResource
- isEmbodiedBy
- isExemplifiedBy
- isMemberOf
- isPartOf
- isRealizedBy
- legalResourceSubdivisionIsPartOf
- treatyProcessHasResultingTreatyDocument
- treatyProcessHasTask

13.4 Concepts

- Abstraction Levels
- Changes
- Citations
- Classified Compilation (CC)
- Dates
- History
- Impacts
- Official Compilation (OC)
- Subdivisions
- Vocabularies

